
Module

 Dateien zur Zusammenfassung von Code zu einem Thema.
 Funktionale Einheit eines VBA-Projekts.
 Wiederverwendung von Code.

Standardmodule

 Verwendung in der prozeduralen Programmierung.
 Dateien mit der Endung *.bas.
 Zusammenfassung von Subroutinen zu einem Thema.
 Deklaration von globalen Variablen, die im gesamten Projekt 

genutzt werden.



Visual Basic for Application Seite 4

Klassenmodule

 Allgemeine Beschreibung von Objekten.
 Attribute und Subroutinen einer Objekt-Kategorie.
 Dateien mit der Endung „.cls“.
 Vordefinierte Module zur Beschreibung von Microsoft Access 

Objekten wie zum Beispiel der Datenzugriff auf Tabellen.
 Benutzerdefinierte Module zur Beschreibung von Objekten wie 

zum Beispiel Kunden etc.

Klasse

 Abstraktion von Dingen aus der realen Welt.
 Allgemeine Beschreibung von einem Microsoft Access Objekt. 

Welche Eigenschaften hat ein Objekt? Wie verhält sich das 
Objekt?

 Bauplan für ein bestimmtes Ding zum Beispiel für ein Formular, 
ein Bericht etc.

 Formale Beschreibung einer Gruppe.
 Vorlage für die Erzeugung eines Objektes.

Klasse „Datensätze (Zeilen) einer Tabelle“

.Fields("Nummer") .Fields("Name") .Fields("Nummer")

.Fields("Nummer") .Fields("Name") .Fields("Nummer")

.Fields("Nummer") .Fields("Name") .Fields("Nummer")

.R
ecordC

ount

.MoveFirst

Attribute

 Beschreibung eines Gegenstandes, einer Person, etc.
 Allgemeingültige Beschreibung für ein Microsoft Access Objekt.
 „Variablen“ eines Objekts.
 Jedes Objekt einer Klasse hat die gleichen Attribute. Zum 

Beispiel hat ein Recordset die Eigenschaft RecordCount (Anzahl 
der Datensätze).

 Die Ausprägung der Attribute kann sich bei den verschiedenen 
Objekten unterscheiden. Zum  Beispiel hat Tabelle A 5 
Datensätze, aber Tabelle B 2 Datensätze.

Beispiel

.Fields("Nummer") .Fields("Name") .Fields("Nummer")

.Fields("Nummer") .Fields("Name") .Fields("Nummer")

.Fields("Nummer") .Fields("Name") .Fields("Nummer")

.R
ecordC

ount

Methoden

 Was kann ein Objekt machen?
 Beschreibung der Funktionalität eines Objekts.
 Veränderungen von Objekten von außen her.
 Prozeduren, die in Klassen gekapselt sind.
 Alle Objekte einer Klasse haben die gleichen Methoden.
 Zum Beispiel wird mit Hilfe von .MoveNext oder .MoveLast der 

Datensatzzeiger einer Tabelle verändert.

Klasse „Recordset“

.Fields("Nummer") .Fields("Name") .Fields("Nummer")

.Fields("Nummer") .Fields("Name") .Fields("Nummer")

.Fields("Nummer") .Fields("Name") .Fields("Nummer")

.MoveFirst.MoveFirst

.MoveLast

Objekte

 Ein Ding (Exemplar, Instanz) aus der realen Welt.
 Objekte wie Formulare und Berichte in Microsoft Access.
 Eine abgeschlossen Einheit.
 Alle Elemente einer Kategorie von Dingen haben die gleichen 

Attribute, aber in unterschiedlichen Ausprägungen. Sie nutzen 
die gleichen Methoden zum Ändern ihrer Attributwerte.

Objekt „Datensätze (Zeilen) der Tabelle ...“

NWTSO-99 Chicken Soup 1,00 €

NWTDFN-14 Walnuts 17,44 €

NWTB-1 Chai 13,50 €

.MoveFirst.MoveFirst

.MoveLast

.R
ecordC

ount

Informationen zu Objekten

 … im VBA-Editor: Ansicht – Objektkatalog.

Aufbau des Objektkatalogs

 Klassen werden in einer Bibliothek wie Bücher 
zusammengefasst. In diesem Beispiel ist die Bibliothek DAO 
ausgewählt.

 In dem linken Listenfeld werden alle Klassen in der 
gewählten Bibiothek angezeigt. In dem Beispiel ist die 
Klasse Recordset ausgewählt. 

 In dem rechten Listenfeld werden alle Elemente einer Klasse 
angezeigt. In diesem Beispiel werden Methoden und 
Attribute aus der Klasse Recordset dargestellt. Die 
Methode .AddNew erzeugt einen neuen Datensatz, das 
Attribut .Bookmark symbolisiert ein Lesezeichen und so 
weiter

Elemente von Klassen im Objektkatalog

Klassen Attribute / Eigenschaften

Standard-Eigenschaften

Methoden

Ereignisse

Methoden

Objektmodell

 Hierarchische Anordnung von Objekten.
 Darstellung von Objekten und deren Abhängigkeit 

untereinander in einem Modell.
 Bibliotheken fassen Objekte in einem Modell zusammen.

Modell „Microsoft Access“

Application
(Access)

Forms
(Formular)

Reports
(Berichte)

Moduls
(Module)

Screen
(Bildschirm)

DoCmd

Controls Controls

Modell „D(ata)A(ccess)O(bject)“ (Ausschnitt)

dbEngine

TableDefs QueryDefs Recordset Relation

Field Index

Workspaces

Databases

Field Parameter Field

Informationen zu Objektmodellen

 … im VBA-Editor: Ansicht – 
Objektkatalog.

 Mit Hilfe des DropDown-Feldes 
am oberen linken Rand wird eine 
Bibliothek ausgewählt. In diesem 
Beispiel ist das Modell DAO 
ausgewählt.

 Entsprechend der gewählten 
Bibliothek werden die 
dazugehörigen Klassen in der 
linken Liste angezeigt.

Elemente im Objektkatalog

Bibliothek

Module

Collections /
Enumeration

Klassen

Konstanten

Eigenschaften

Standard-Eigenschaften

Methoden

Ereignisse

Methoden

Methoden

Konstanten

Verweise auf Bibliotheken

 … im VBA-Editor: Extras – Verweise.

Standardverweise

 Visual Basic for Application. Die Bibliothek beschreibt alle 
Elemente der Programmiersprache VBA.

 OLE Automation. Einbetten oder Verlinken von Objekten wie 
zum Beispiel Bilder.

 Microsoft Access 14.0 Object Library. Beschreibung der 
Objekte von Microsoft Access. Die Version 14.0 bezieht sich 
auf Microsoft Access 2010.

 Microsoft Office 14.0 Access database engine Object Library. 
Beschreibung der Datenbank selbst und deren Datenzugriff 
darauf. Die Bibliothek benutzt die Jet-Datenbank Version 4.0 
(.mdb) oder die Version 12.0 (accdb).

Einfügung von Verweisen

 Extras – Verweise.
 In der Liste wird zum Beispiel zu dem Eintrag Microsoft DAO 

3.6 Object Library geblättert. Hinweis: Die Liste ist alphabetisch 
sortiert.

 Klick in das Kontrollkästchen links vom Namen des Verweises. 
Das Häkchen in einem Kästchen kennzeichnet den aktiven 
Verweis.

 OK schließt das Fenster.

D(ata)A(ccess)O(bject)

 Arbeiten mit der Datenbank.
 Datenzugriff auf Tabellen und Abfragen in Microsoft Access.
 Die Bibliothek Microsoft DAO 3.6 Object Library muss für 

Microsoft Access 2007 und älter eingebunden werden.
 Seit Microsoft Access 2010  Standardbibliothek unter dem 

Namen Microsoft Office 14.0 Access database engine Object 
Library. 

Objekt-Variable

 Die Variable speichert einen Verweis auf ein Objekt vom Typ 
„Klasse“.

 Instanzen von einer Klasse aus einer bestimmten Bibliothek.
 Platzhalter für eine Referenz auf ein konkretes Objekt. 

Sub datenzugriff()

    Dim dbs As DAO.Database
    Dim rs As DAO.Recordset

End Sub

… deklarieren

 Der Variablenname ist frei wählbar. Der gewählte Name sollte 
aber das zu speichernde Objekt widerspiegeln.

 Dem Schlüsselwort As folgt die Klasse. Die Klasse spiegelt den 
gewünschten Objekttyp wieder.

Dim Asdbs DAO . Database

Dim As[name] [klasse][bibliothek] .

… von der Klasse

 As folgt immer ein Klassenname. In diesem Beispiel Database.
 Der Klassenname gibt über die Objektkategorie Auskunft.

Dim Asdbs DAO . Database

Dim As[name] [klasse][bibliothek] .

Dim Asdbs Database

… in der Bibliothek

 Der Klassenname kann mit einem Bibliotheksnamen verbunden 
werden. 

 Der Punkt-Operator verbindet den Bibliotheksnamen, in dem 
die gewünschte Klasse definiert ist, mit dem Klassennamen. 

 Der Bibliotheksname kann weggelassen werden, wenn die 
Klasse eindeutig einer Bibliothek zugeordnet werden kann.

Dim Asdbs DAO . Database

Dim As[name] [klasse][bibliothek] .

Initialisierung

 Mit Hilfe des Gleichheitszeichens wird der Variablen eine 
Referenz auf ein bestimmtes Objekt zugewiesen.

 Die Objekt-Variable verweist auf ein konkretes Objekt.

    Dim dbs As DAO.Database
    
    Set dbs = CurrentDb

Erläuterung

 Die Zuweisung beginnt mit dem Schlüsselwort Set (setze).
 Dem Schlüsselwort folgt der Name.
 Mit Hilfe des Gleichheitszeichens wird der Variablen ein 

Verweis auf ein Objekt übergeben. In diesem Beispiel wird ein 
Verweis auf die aktuelle Datenbank (CurrentDB) übergeben.

Set =dbs CurrentDB

Set =[name] [objekt]

Objekt-Variable leeren

 Das Schlüsselwort Nothing ist ein Platzhalter für „kein Verweis“.
 Die Objekt-Variable verweist auf kein Objekt.

    Dim dbs As DAO.Database
    
    Set dbs = Application.CurrentDb
    Set dbs = Nothing

Objekt-Variable vom Typ „Datenbank“

 Die Objekt-Variable kann auf Objekte von der Klasse 
DAO.Database verweisen.

 Die Objekt-Variable kann auf eine Microsoft Access-Datenbank 
oder eine ODBC-Datenbank verweisen.

    Dim dbs As DAO.Database

Verweis auf die aktuell geöffnete Datenbank

 Die Methode CurrentDb der Klasse Application gibt ein Verweis 
auf die aktuelle Datenbank zurück.

 Der Verweis auf die aktuelle Datenbank wird in diesem Beispiel 
in der Objekt-Variable dbs gespeichert.

    Dim dbs As DAO.Database
    
    Set dbs = Application.CurrentDb

Klassen und Methoden

 Die Methode CurrentDb der Klasse Application gibt ein Verweis 
auf die aktuelle Datenbank zurück.

 Klassen und die darin definierten Methoden werden mit dem 
Punktoperator verbunden.

 Wenn die Methode eindeutig einer Klasse zugeordnet werden 
kann, kann der Klassenname entfernt werden.

    Dim dbs As DAO.Database
    
    Set dbs = Application.CurrentDb
    Set dbs = CurrentDb

Speicherort der aktuellen Datenbank

 Das Wurzelelement Application (Microsoft Access) hat das 
Objekt CurrentProject (das aktuelle Projekt).

 Das Objekt CurrentProject hat das Attribut Path.
 Die Eigenschaft Path gibt den Pfad des aktuell geöffneten 

Projekts (der Microsoft Access - Datei) wieder.

    Dim speicherort As String
    
    speicherort = Application.CurrentProject.Path

Klassen und Attribute

 Klassen und die darin definierten Attribute werden mit dem 
Punktoperator verbunden.

    Dim speicherort As String
    
    speicherort = Application.CurrentProject.Path

Abbildung des Objektmodells

 Mit Hilfe des Punktoperators wird die Hierachie in einem 
Objektmodell abgebildet.

 In diesem Beispiel wird das Nachfolger-Element CurrentProject 
dem Vorgänger-Element Application zugeordnet.

 Die erste Klasse in der Auflistung spiegelt die Wurzel des 
Objektmodells wieder. In diesem Beispiel ist die Klasse 
Application die Wurzel des Objektmodells „Microsoft Access“. 

    Dim speicherort As String
    
    speicherort = Application.CurrentProject.Path

Verweis auf eine externe Datenbank

    Dim wks As DAO.Workspace
    Dim dbs As DAO.Database
    Dim speicherort As String
    
    speicherort = Application.CurrentProject.Path 
    speicherort = speicherort & "\" & "vba04_Beispiel_Backend.accdb"
    
    Set wks = DBEngine.Workspaces(0)
    Set dbs = wks.OpenDatabase(speicherort)
    
    dbs.Close
    Set dbs = Nothing
    Set wks = Nothing

Schritte 

 Zuerst wird ein Arbeitsbereich für die zu öffnende Datenbank 
definiert.

 Im nächsten Schritt wird die Datenbank geöffnet.
 Falls die Datenbank nicht mehr gebraucht wird, wird die 

Datenbank geschlossen.

1. Schritt: Definition eines Arbeitsbereichs

 Die Variable wks kann einen Verweis auf einen Arbeitsbereich / 
Benutzersitzung (DAO.Workspace) speichern.

 Der Ausdruck DBEngine.Workspaces(0) ist ein Verweis auf den 
Standard-Arbeitsbereich von Microsoft Access. Der 
Benutzername im Standard-Arbeitsbereich lautet „Admin“.

    Dim wks As DAO.Workspace
    Set wks = DBEngine.Workspaces(0)
    Set wks = Nothing

dbEngine

 Eigenschaft der Application (der Microsoft Access Anwendung).
 Wurzel-Element des Modells „DAO“.
 Platzhalter für die Datenbank.

Set =wks dbEngine

Arbeitsbereich

 Die Hierachie der Objekte wird durch den Punkt symbolisiert.
 Das Objekt Workspaces ist ein Kind-Element von dbEngine.
 Der Buchstabe „s“ am Ende des Namens kennzeichnet eine 

Sammlung (Collection). Sammlungen sind null-basiert. Die 0 in 
den runden Klammern symbolisiert das erste Element in der 
Sammlung.

Set =wks )Workspaces.dbEngine ( 0

2. Schritt: Öffnen der Datenbank

 Die Variable dbs kann einen Verweis auf eine Datenbank 
(DAO.Database) speichern.

 Die Methode .OpenDatabase ist in DAO.Workspace definiert und 
öffnet eine Datenbank in dem Standard-Arbeitsbereich.

 In den runden Klammern wird der Methode der Name  
Datenbank übergeben. Der Pfad zu der Datenbank muss 
absolut angegeben werden.

    Dim dbs As DAO.Database
    
    Set dbs = wks.OpenDatabase(speicherort)

3. Schritt: Schließen der Datenbank 

 Mit Hilfe der Methode .close wird eine, von VBA geöffnete 
Datenbank geschlossen.

 Falls die aktuelle Datenbank mit Hilfe von Application.CurrentDb 
geöffnet wird, wird die Methode nicht benötigt.

    Dim dbs As DAO.Database
    
    Set dbs = wks.OpenDatabase(pfad)
    dbs.close

Objekt-Variablen und Methoden

 Objekt-Variablen und Methoden werden mit einem Punkt 
verbunden.

 Die Methode rechts vom Punkt verändert die Instanz links vom 
Punkt.

 Die Objekt-Variable ist vom Typ einer Klasse … In dieser 
Klasse ist Methode (rechts vom Punkt) definiert.

 Der Methodenname ist in der Klasse eindeutig.
 Dem Methodenname folgt eine Parameterliste.

Set =dbs )OpenDatabase.wks ( pfad

Close.dbs

Parameterliste von Methoden

 Dem Methodenname folgt eine Parameterliste.
 In der Parameterliste werden der Methode Parameter als Start-

Werte für in der Methode definierte Variablen übergeben. Der 
Methode OpenDatebase wird der Name und Speicherort der zu 
öffnenden Datenbank übergeben.

 Die Parameter werden durch Kommata getrennt.
 Die Anzahl der Parameter sind abhängig von dem definierten 

Methodenkopf.

Set =dbs )OpenDatabase.wks ( pfad

Close.dbs

Leere Parameterliste von Methoden

 Die Klammern können leer sein. Der Methode werden keine 
Parameter übergeben.

 Die Klammern der Parameterliste können weggelassen 
werden.

Set =dbs )OpenDatabase.wks ( pfad

Close.dbs

Objekt-Variable vom Typ „Recordset“

 Die Objekt-Variable kann auf Objekte von der Klasse 
DAO.Recordset verweisen.

 Die Objekt-Variable kann auf Datensätze (Zeilen) in Tabellen 
und Abfragen verweisen.

Dim rs As DAO.Recordset

Öffnen eines Recordsets

 Mit Hilfe der Methode .OpenRecordset() werden Datensätze zum 
Lesen und Schreiben geöffnet.

 Falls das zu öffnende Element nicht vorhanden ist, wird ein 
Laufzeitfehler angezeigt.

Set rs = dbs.OpenRecordset(tblName, dbOpenTable)
    
rs.Close

Erläuterung

 Direkt auf den Methodennamen folgt die Parameterliste.
 Die einzelnen Parameter werden durch Kommata getrennt.
 Erster Parameter: Wo sind die Datensätze gespeichert? In 

diesem Beispiel wird ein Tabellenname genutzt.
 Zweiter Parameter: Wie erfolgt der Zugriff? Mit Hilfe von 

Konstanten wird der Zugriff angegeben. In diesem Beispiel wird 
einen Tabelle geöffnet. Der Nutzer hat Lese- und 
Schreibrechte.

Set =rs )OpenRecordset.dbs ( tblName , dbOpenTable

Möglichkeiten für den Zugriff

 dbOpenTable. Standardzugriff, wenn keine Angaben gemacht 
wurden. Öffnen einer lokalen Tabelle.

 dbOpenDynaset. Öffnen einer verknüpften Tabelle oder einer 
Abfrage. Die Abfrage kann in Microsoft Access definiert sein 
oder direkt mit Hilfe von SQL angegeben werden.

 dbOpenSnapshot. „Momentaufnahme“ von Datensätzen. Nur 
lesenden Zugriff.

Schließen eines Recordsets

 Sobald die Datensätze nicht mehr benötigt werden, wird die 
Tabelle mit der Methode .close geschlossen.

rs.Close

Anzahl der Datensätze

 Hinweis: Die Datensätze wurden mit Hilfe der Konstanten 
dbOpenTable geöffnet.

 Das  Attribut .RecordCount enthält die Anzahl der Datensätze 
der geöffneten Tabelle.

 Debug.Print "Anzahl der Datensätze: " & rs.RecordCount



Visual Basic for Application Seite 54

Navigation in den Datensätzen

BOF (begin of)

Datensatz 1

Datensatz 2

Datensatz 3

Datensatz 4

EOF (end of)

Datensatzzeiger

rs.MovePrevious

rs.MoveNext

rs.MoveLast

rs.MoveFirst

Felder in einem Datensatz

 Jeder Datensatz hat eine bestimmte Anzahl von Datenfeldern.
 Die Datenfelder können mit Hilfe der Sammlung .Fields eines 

Recordset durchlaufen werden.
 Die Sammlung .Fields enthält alle Feldnamen (siehe 

Entwurfsansicht einer Tabelle)

produktnummer = rs.Fields("Produktcode")
produktnummer = rs.Fields(3)

… mit Hilfe des Index lesen

 .Fields ist eine Sammlung von Datenfeldern.
 Dem Namen der Sammlung folgen direkt die runden Klammern. 

In den runden Klammern wird ein Index angegeben.
 Ein Index einer Sammlung ist immer eine Ganzzahl.
 Auflistungen sind null-basiert. Das erste Element hat den Index 

0, das zweite Element 1 und so weiter.
 Falls das Element in der Sammlung nicht vorkommt, wird der 

Fehler „Element in der Auflistung nicht gefunden“ angezeigt.

produktnummer = rs.Fields(3)



Visual Basic for Application Seite 57

… mit Hilfe des Namens lesen

 Jedes Element in der Sammlung .Fields kann über den Namen 
angesprochen werden.

 Der Name des Datenfeldes wird in den runden Klammern der 
Sammlung als String übergeben.

 Der Name kennzeichnet eindeutig ein Datenfeld in dem 
geöffneten Datensatz.

produktnummer = rs.Fields("Produktcode")

Weitere Schreibweise

 Die Objekt-Variable wird vom Namen des Datenfeldes durch 
ein Ausrufezeichen getrennt.

 Falls der benutzerdefinierte Name Sonderzeichen wie zum 
Beispiel das Leerzeichen oder Umlaute enthält, muss der 
Name mit Hilfe der eckigen Klammern zusammengefasst 
werden.

produktnummer = rs.Fields("Produktcode")
produktnummer = rs!Produktcode

stueckzahlMin = rs![Mindeststückzahl für Nachbestellung]

Punkt-Operator

 Rechts vom Punkt-Operator befindet sich ein, in Microsoft 
Office definiertes Element.

 Objekt-Variablen und deren Eigenschaften und Methoden 
werden verbunden.

 Die Hierarchie in einem Objektmodell wird abgebildet.
 Bibliotheken und deren darin enthaltenen Klassen werden 

verknüpft.

Nutzung des Ausrufezeichens

 Verbindung von Objekten und deren Kind-Elementen.
 Rechts vom Ausrufezeichen werden benutzerdefinierte Namen 

von Datenfelder, Objekten etc. genutzt.

Wert eines Datenfeldes

 Die Eigenschaft .value gibt den Wert des Datenfeldes zurück.
 Es wird der, in einem Datenfeld gespeicherte Wert gelesen.

produktnummer = rs.Fields("Produktcode").Value
produktnummer = rs.Fields(3).Value
produktnummer = rs!Produktcode.Value

Hinzufügung eines Datensatzes

 Datensätze werden immer am Ende einer Tabelle eingefügt.
 Datenfelder vom Typ „AutoWert“ werden automatisiert gesetzt.

    rs.AddNew

    rs.Fields("Produktcode").Value = "NWTMN-1"
    rs.Fields("Artikelname").Value = "Neuer Artikel"
    
    rs.Update

1. Schritt: Datensatz anlegen

 Der Datensatz wird am Ende der Tabelle neu angelegt.
 Die Datenfelder werden mit den angegebenen Standardwerten 

belegt.

 rs.AddNew

2. Schritt: Datenfelder füllen

 Links vom Zuweisungsoperator steht das zu füllende Datenfeld. 
 Das Datenfeld ist ein Synonym für eine Variable.

 Rechts vom Zuweisungsoperator steht der zu speichernde 
Wert. Der Wert kann als Literal im Code übergeben werden. 
Der Wert kann aber auch durch einen Ausdruck berechnet 
werden. 

    rs.AddNew

    rs.Fields("Produktcode").Value = "NWTMN-1"
    rs.Fields("Artikelname").Value = "Neuer Artikel"

3. Schritt: Änderungen speichern

 rs.Update speichert den neuen Datensatz.
 Der Datensatzzeiger zeigt auf den eingefügten Datensatz.

    rs.AddNew

    rs.Fields("Produktcode").Value = "NWTMN-1"
    rs.Fields("Artikelname").Value = "Neuer Artikel"
    
    rs.Update

Änderungen an einem Datensatz

 Mit Hilfe den entsprechenden Befehlen wird der 
Datensatzzeiger auf den zu ändernden Datensatz verschoben.

    rs.Edit

    rs.Fields("Beschreibung").Value = "Beschreibungstext"
    
    rs.Update

1. Schritt: Kopieren des Datensatzes

 Der Datensatz wird zur Bearbeitung in einem Puffer 
gespeichert.

    rs.Edit

2. Schritt: Änderungen des Wertes

 Das zu ändernde Datenfeld steht links vom 
Zuweisungsoperator.

 Rechts vom Zuweisungsoperator steht der zu speichernde 
Wert. Der Wert kann als Literal im Code übergeben werden. 
Der Wert kann aber auch durch einen Ausdruck berechnet 
werden. 

    rs.Edit

    rs.Fields("Beschreibung").Value = "Beschreibungstext"

3. Schritt: Änderungen speichern

 rs.Update speichert den neuen Datensatz. Falls die Methode 
nicht aufgerufen wird, werden die Änderungen verworfen.

 Der Datensatzzeiger zeigt auf den geänderten Datensatz.

    rs.Edit

    rs.Fields("Beschreibung").Value = "Beschreibungstext"
    
    rs.Update

Löschung eines Datensatzes

 Mit Hilfe den entsprechenden Befehlen wird der 
Datensatzzeiger auf den zu löschenden Datensatz verschoben.

 Die Methode .Delete löscht den aktuellen Datensatz ohne 
Nachfrage.

 Alle nachfolgenden Datensätze werden um eine Position nach 
oben verschoben. 

 Der Datensatzzeiger zeigt aber auf die gleiche Position wie vor 
der Löschung, aber auf einen anderen Datensatz.

    rs.MoveLast

    rs.Delete

Datensatz-Zeiger vor der Löschung

BOF (begin of)

Datensatz 1

Datensatz 2

Datensatz 3

Datensatz 4

EOF (end of)

Datensatz-Zeiger nach der Löschung

BOF (begin of)

Datensatz 1

Datensatz 3

Datensatz 4

EOF (end of)

Objekt-Variable „Tabellendefinition“

 Die Variable vom Datentyp DAO.TableDef speichert einen 
Verweis auf die Struktur einer Tabelle.

 Speicherung auf die Referenz einer Tabellendefinition.
 Die Struktur einer Tabelle wird in der Entwurfsansicht einer 

Microsoft Access – Tabelle abgebildet.

    Dim dbs As DAO.Database
    Dim tblDef As DAO.TableDef

Tabellenstruktur lesen

 Mit Hilfe von Application.CurrentDb.TableDefs("Artikel") wird der 
Objekt-Variablen tblDef ein Verweis auf die Struktur der Tabelle 
Artikel übergeben.

    Dim dbs As DAO.Database
    Dim tblDef As DAO.TableDef
    
    Set dbs = Application.CurrentDb
    Set tblDef = dbs.TableDefs("Artikel")

Erläuterung

 Die Objekt-Variable dbs ist ein Platzhalter für eine Datenbank.
 Jede Klasse vom Objekt Database besitzt eine Sammlung 

(Collection) von Tabellendefinition.
 Die Sammlung .TableDefs enthält zu jeder Tabelle in der 

gewählten Datenbank einen Eintrag.
 In den runden Klammern wird der Sammlung der Name der 

Tabelle als Index übergeben.
 Die Tabellendefinition des gewählten Elements wird als 

Verweis in der Objekt-Variablen gespeichert.

Neue Tabellenstruktur anlegen

 Mit Hilfe von .CreateTableDef("KundenPlusBestellung") wird eine 
neue leere Tabellenstruktur erzeugt.

 Der Methode wird der Name der zu erzeugenden Tabelle in 
den runden Klammern übergeben.

 Der VBA-Befehl entspricht Erstellen – Tabellenentwurf in der 
Gruppe Tabellen.

    Dim dbs As DAO.Database
    Dim tblDef As DAO.TableDef
    
    Set dbs = Application.CurrentDb
    Set tblDef = dbs.CreateTableDef("KundenPlusBestellung")

… und in die Datenbank einfügen

 Mit Hilfe von .Append wird der Sammlung TableDefs ein Element 
hinzugefügt.

 Hinweis: Die Tabelle muss mindestens ein Feld enthalten.
 Die Methode .RefreshDatabaseWindow des Objekts Application 

aktualisiert das Datenbankfenster. Die neu erstellte Tabelle 
wird im Navigationsfenster angezeigt.

    dbs.TableDefs.Append tblDef
     
    Application.RefreshDatabaseWindow

Alle Felder in einer Tabelle

 Die Objektvariable tblDef hat ein untergeordnetes Objekt 
.Fields.

 Das untergeordnete Objekt ist eine Auflistung aller Felder in der 
angegebenen Tabelle.

 Die Definition der Tabellenspalten kann gelesen und verändert 
werden.

    Dim tblDef As DAO.TableDef
    
    Set dbs = Application.CurrentDb
    Set tblDef = dbs.TableDefs("Artikel")

    Debug.Print tblDef.Fields(0).Name
    Debug.Print tblDef.Fields(1).Name

Anzahl der Felder

 Das Attribut .Count gibt die Anzahl der Felder in der 
angegebenen Tabelle zurück.

    Dim tblDef As DAO.TableDef
    
    Set dbs = Application.CurrentDb
    Set tblDef = dbs.TableDefs("Artikel")

    Debug.Print tblDef.Fields.Count

Feld x in der Sammlung

 Dem Namen der Sammlung .Fields folgt in runden Klammern 
ein Index.

 Mit Hilfe des Index wird eindeutig ein Element in der Sammlung 
identifiziert.

 Der Index ist immer vom Datentyp „Ganzzahl“.

    Dim tblDef As DAO.TableDef
    
    Set dbs = Application.CurrentDb
    Set tblDef = dbs.TableDefs("Artikel")

    Debug.Print tblDef.Fields(0).Name
    Debug.Print tblDef.Fields(1).Name

Index eines Elements

 Als Index wird immer eine Ganzzahl genutzt.
 Das erste Element hat den Index 0.
 Das letzte Element hat den Index tblDef.Fields.Count – 1.

    Dim tblDef As DAO.TableDef
    
    Set dbs = Application.CurrentDb
    Set tblDef = dbs.TableDefs("Artikel")

    Debug.Print tblDef.Fields(0).Name
    Debug.Print tblDef.Fields(1).Name

Lesen der Feldnamen

 Das Attribut .Name gibt den Namen des Elements in der 
Auflistung wieder.

 In diesem Beispiel wird der der erste und zweite Feldname 
einer Tabelle zurückgegeben.

    Dim tblDef As DAO.TableDef
    
    Set dbs = Application.CurrentDb
    Set tblDef = dbs.TableDefs("Artikel")

    Debug.Print tblDef.Fields(0).Name
    Debug.Print tblDef.Fields(1).Name

Objekt-Variable vom Typ „Tabellenfeld“

 Die Bibliothek .DAO enthält die Klasse .Field.
 Die Klasse ist Basis für Tabellenfelder einer Tabelle in 

Microsoft Access.

    Dim fld As DAO.Field



Visual Basic for Application Seite 84

Neues Feld anlegen

 Mit Hilfe der Methode .CreateField() eines Objekts 
„Tabellendefinition“ wird ein neues Feld erzeugt.

 Der Methode wird als erster Parameter der Name des Feldes 
übergeben.

    Dim fld As DAO.Field
    
    Set tblDef = dbs.CreateTableDef("KundenPlusBestellung")
    
    Set fld = tblDef.CreateField("KundenName", dbText, 255)

Parameter der Methode .CreateField()

 Der Methode wird als erster Parameter der Name des Feldes 
übergeben.

 Der zweite Parameter legt den Datentyp des Feldes fest. In 
diesem Beispiel wird ein Feld vom Datentyp „Text“ und ein 
weiteres Feld vom Datentyp „Long“ angelegt. Eine Auflistung 
der Datentypen finden Sie unter https://msdn.microsoft.com/de-
de/library/office/ff845405.aspx.

    Dim fld As DAO.Field
    
    Set fld = tblDef.CreateField("KundenName", dbText, 255)
    Set fld = tblDef.CreateField("IDBestellsumme", dbLong)

Erzeugung von Feldern vom Typ „Text“

 Mit Hilfe der Konstanten dbText wird ein Feld vom Datentyp 
„Text“ angelegt.

 Der dritte Parameter legt die maximale Zeichenlänge fest.

    Dim fld As DAO.Field
    
    Set fld = tblDef.CreateField("KundenName", dbText, 255)

Felder in die Sammlung „Fields“ einhängen

 Die Methode .Append fügt in die Sammlung Fields ein 
passendes Objekt ein.

 Die Sammlung .Fields des Platzhalter tblDef wird verändert. Die 
Objekt-Variable tblDef verweist auf die zu ändernde 
Tabellendefinition.

    Dim fld As DAO.Field
    
    Set fld = tblDef.CreateField("KundenName", dbText, 255)
    tblDef.Fields.Append fld

Feld vom Datentyp „AutoInkrement“

 Das Feld ist vom Datentyp dbLong. Das Feld kann Ganzzahlen 
vom Datentyp Long speichern.

 Jedes Feld hat Attribute (Attributes). Mit Hilfe von Konstanten 
können die Attribute gesetzt werden.

 In diesem Beispiel wird bei Neuanlage des Datensatz das Feld 
automatisch auf einen eindeutigen Wert vom Typ Long gesetzt. 
Ein interner Zähler wird um eins hochgezählt-

Set fld = tblDef.CreateField("IDBestellsumme", dbLong)
fld.Attributes = dbAutoIncrField
tblDef.Fields.Append fld

Objekt-Variable vom Typ „Index“

 Die Bibliothek .DAO enthält die Klasse .Index.
 Jedes Feld in einer Datenbank kann indexiert werden. Die 

Suche nach Werten wird durch einen Index beschleunigt.

   Dim idx As DAO.Index

Erzeugung eines Indizes

 Mit Hilfe der Methode .CreateIndex wird ein neuer Index erzeugt.
 Der Methode wird in runden Klammern der eindeutige Name 

des Index übergeben.
 Der Objekt-Variablen wird ein Verweis auf den neu erzeugten 

Index übergeben.

    Dim dbs As DAO.Database
    Dim tblDef As DAO.TableDef
    Dim idx As DAO.Index
    
    Set dbs = Application.CurrentDb
    Set tblDef = dbs.TableDefs("KundenPlusBestellung")
    
    Set idx = tblDef.CreateIndex("PrimaryKey")

Alle Indizies einer Tabelle

 Die Sammlung .Indexes eines Objekts „Tabellendefinition“ 
enthält alle Indizies.

    Dim tblDef As DAO.TableDef
    Dim idx As DAO.Index
    
    Set tblDef = dbs.TableDefs("KundenPlusBestellung")
    
    Set idx = tblDef.CreateIndex("PrimaryKey")
    tblDef.Indexes.Append idx

Anhängen eines neuen Index

 Mit Hilfe der Methode .Append einer Sammlung wird der 
Auflistung .Indexes ein neuer Index hinzugefügt.

 In diesem Beispiel wird der Index PrimaryKey hinzugefügt.

    Dim tblDef As DAO.TableDef
    Dim idx As DAO.Index
    
    Set tblDef = dbs.TableDefs("KundenPlusBestellung")
    
    Set idx = tblDef.CreateIndex("PrimaryKey")
    tblDef.Indexes.Append idx

Zusammenfassung von Anweisungen

 
    With idx

        Set fld = .CreateField("IDBestellsumme")
        .Fields.Append fld
        .Unique = False
        .Primary = True

    End With

Aufbau

With [objektvariable]

End With

[anweisung]

Erläuterung

 Zusammenfassung von Anweisungen, die ein Objekt 
verändern.

 Dem Schlüsselwort With folgt das Objekt oder ein Platzhalter 
für das Objekt.

 Zwischen den Schlüsselwörtern With und End With folgen 
Anweisungen, die das Objekt verändern oder Methoden des 
Objekts nutzen.

Aufbau der Anweisungen

Set fld

End With

.CreateField("IDBestellsumme")= .

With idex

Set fld .CreateField("IDBestellsumme")= .idex

Erläuterung

 Dem Schlüsselwort With folgt ein Platzhalter für ein Objekt oder 
das Objekt selbst.

 Methoden und Attribute, die sich auf dieses Objekt beziehen, 
benötigen links vom Punktoperator keine Angabe des Objekts. 

 Der Punktoperator zur Verknüpfung mit dem angegebenen 
Objekt wird aber benötigt. 

 Die genutzten Methoden oder Attribute können rechts oder 
links vom Zuweisungsoperator stehen.

Eindeutiger Index?

 Die Eigenschaft .Unique legt fest, ob der Index eindeutig ist 
oder nicht.

    Dim fld As DAO.Field
    Dim idx As DAO.Index
    
    Set idx = tblDef.CreateIndex("PrimaryKey")
    idex.Unique = False

Setzen eines Primärschlüssels

 Die Eigenschaft .Primary legt fest, ob der Index als 
Primärschlüssel genutzt wird oder nicht.

 Ein Primärschlüssel ist für jeden Datensatz eindeutig.

    Dim fld As DAO.Field
    Dim idx As DAO.Index
    
    Set idx = tblDef.CreateIndex("PrimaryKey")
    idex.Primary = True

Erzeugung von Feldern für ein Index

 Mit Hilfe der Methode .CreateField() der Klasse „Index“ wird ein 
Feld im Index erzeugt.

 Setzen der Feldeigenschaft Indiziert eines Feldes in der 
Entwurfsansicht eines Feldes.

 Der Methode wird der Feldname übergeben.
 Hinweis: Das Feld muss in der Tabelle existieren.

    Dim fld As DAO.Field
    Dim idx As DAO.Index
    
    Set idx = tblDef.CreateIndex("PrimaryKey")
   
    Set fld = idex.CreateField("IDBestellsumme")

… und anhängen

 Der Sammlung .Fields der Klasse „Index“ wird mit Hilfe der 
Methode .Append das neu erzeugte Feld angehängt.

    Dim fld As DAO.Field
    Dim idx As DAO.Index
    
    Set idx = tblDef.CreateIndex("PrimaryKey")
   
    Set fld = idex.CreateField("IDBestellsumme")
    idex.Fields.Append fld

Attribute VB_Name = "ZugriffDatabase"
Option Compare Database
Option Explicit

Sub datenzugriff_DatenbankAktuell()
    Dim dbs As DAO.Database
    
    Set dbs = Application.CurrentDb
    
    Set dbs = Nothing
End Sub



Attribute VB_Name = "ZugriffDatabase"
Option Compare Database
Option Explicit

Sub datenzugriff_Backend()
    Dim wks As DAO.Workspace
    Dim dbs As DAO.Database
    Dim rs As DAO.Recordset
    
    Dim tblName As String
    Dim speicherort As String
    Dim produktnummer As String
    Dim produktname As String
    Dim produktbeschreibung As String
    
    speicherort = Application.CurrentProject.Path & "\" & "vba04_Beispiel_Backend.accdb"
    
    Set wks = DBEngine.Workspaces(0)
    Set dbs = wks.OpenDatabase(speicherort)
    
    tblName = "Artikel"
    Set rs = dbs.OpenRecordset(tblName, dbOpenTable)
    
    rs.MoveFirst
    
    produktnummer = rs.Fields("Produktcode")
    Debug.Print produktnummer
    produktname = rs.Fields("Artikelname").Value
    Debug.Print produktname
    
    rs.Close
    Set rs = Nothing
    
    dbs.Close
    Set dbs = Nothing
End Sub



Attribute VB_Name = "ZugriffFields"
Option Compare Database
Option Explicit

Sub datenzugriff_Datenfelder()
    Dim dbs As DAO.Database
    Dim rs As DAO.Recordset
    
    Dim tblName As String
    Dim produktnummer As String
    Dim produktname As String
    Dim auslaufartikel As Boolean
    Dim stueckzahlMin As Integer
    Dim mindestbestand As Integer
    
    tblName = "Artikel"
    Set dbs = Application.CurrentDb
    
    Set rs = dbs.OpenRecordset(tblName, dbOpenTable)
    
    rs.MoveFirst
    
    produktnummer = rs.Fields("Produktcode")
    Debug.Print produktnummer
    produktname = rs.Fields("Artikelname").Value
    Debug.Print produktname
    
    auslaufartikel = rs!auslaufartikel
    Debug.Print auslaufartikel
    stueckzahlMin = rs![Mindeststückzahl für Nachbestellung].Value
    Debug.Print stueckzahlMin

    mindestbestand = rs.Fields(7).Value
    Debug.Print mindestbestand
    
    rs.Close
    Set rs = Nothing
    
    Set dbs = Nothing
End Sub

Sub datensatz_Loeschen()
    Dim dbs As DAO.Database
    Dim rs As DAO.Recordset
    
    Dim tblName As String
    Dim produktnummer As String
    Dim produktname As String
    Dim produktbeschreibung As String
    Dim intKategorie As Integer
    
    tblName = "Artikel"
    Set dbs = Application.CurrentDb
    
    Set rs = dbs.OpenRecordset(tblName, dbOpenTable)
    
    rs.MoveLast
    
    rs.Delete
    
    rs.Close
    Set rs = Nothing
    
    Set dbs = Nothing
End Sub

Sub datensatz_Aendern()
    Dim dbs As DAO.Database
    Dim rs As DAO.Recordset
    
    Dim tblName As String
    Dim produktnummer As String
    Dim produktname As String
    Dim produktbeschreibung As String
    Dim intKategorie As Integer
    
    tblName = "Artikel"
    Set dbs = Application.CurrentDb
    
    Set rs = dbs.OpenRecordset(tblName, dbOpenTable)
    
    rs.MoveLast
    
    rs.Edit
    rs.Fields("Beschreibung").Value = "Beschreibungstext"
    rs.Update
    
    rs.Close
    Set rs = Nothing
    
    Set dbs = Nothing
End Sub



Sub datensatz_Neu()
    Dim dbs As DAO.Database
    Dim rs As DAO.Recordset
    
    Dim tblName As String
    Dim produktnummer As String
    Dim produktname As String
    Dim produktbeschreibung As String
    Dim intKategorie As Integer
    
    tblName = "Artikel"
    Set dbs = Application.CurrentDb
    
    Set rs = dbs.OpenRecordset(tblName, dbOpenTable)
    
    rs.MoveFirst
    
    rs.AddNew
    rs.Fields("Produktcode").Value = "NWTMN-1"
    rs.Fields("Artikelname").Value = "Neuer Artikel"
    
    rs.Update
    
    rs.Close
    Set rs = Nothing
    
    Set dbs = Nothing
End Sub



Attribute VB_Name = "ZugriffRecordset"
Option Compare Database
Option Explicit

Sub datenzugriff_Tabelle()
    Dim dbs As DAO.Database
    Dim rs As DAO.Recordset
    Dim tblName As String
    
    tblName = "Artikel"
    Set dbs = Application.CurrentDb
    
    Set rs = dbs.OpenRecordset(tblName, dbOpenTable)
    
    rs.Close
    Set rs = Nothing
    
    Set dbs = Nothing
End Sub

Sub AnzahlDatensaetze()
    Dim dbs As DAO.Database
    Dim rs As DAO.Recordset
    Dim tblName As String
    
    tblName = "Artikel"
    Set dbs = Application.CurrentDb
    
    Set rs = dbs.OpenRecordset(tblName, dbOpenTable)
    Debug.Print "Anzahl der Datensätze: " & rs.RecordCount
    
    rs.Close
    Set rs = Nothing
    
    Set dbs = Nothing
End Sub



Attribute VB_Name = "ZugriffTableDef"
Option Compare Database
Option Explicit

Sub TabelleEntwurf()
    Dim dbs As DAO.Database
    Dim tblDef As DAO.TableDef
    Dim fld As Field
    Dim zaehler As Integer
    
    Set dbs = Application.CurrentDb
    Set tblDef = dbs.TableDefs("Artikel")
    
    Debug.Print tblDef.Fields.Count
    Debug.Print tblDef.Fields(0).Name
    Debug.Print tblDef.Fields(1).Name
    Debug.Print tblDef.Fields(2).Name
     
    Set fld = tblDef.CreateField(fldName, dbText, textLaenge)
    
    tblDef.Fields.Append fld
    
    Set tblDef = Nothing
    
    Set dbs = Nothing
End Sub

Sub TabelleAnlegen()
    Dim dbs As DAO.Database
    Dim tblDef As DAO.TableDef
    Dim fld As DAO.Field
    
    Set dbs = Application.CurrentDb
    Set tblDef = dbs.CreateTableDef("KundenPlusBestellung")
    
    Set fld = tblDef.CreateField("KundenName", dbText, 255)
    tblDef.Fields.Append fld
    
    Set fld = tblDef.CreateField("Bestellnummer", dbText, 50)
    tblDef.Fields.Append fld
    
    Set fld = tblDef.CreateField("IDBestellsumme", dbLong)
    fld.Attributes = dbAutoIncrField
    tblDef.Fields.Append fld
    
    dbs.TableDefs.Append tblDef
    
    Set fld = Nothing
    Set tblDef = Nothing
    Set dbs = Nothing
    
    Application.RefreshDatabaseWindow
End Sub


Sub IndexAnlegen()
    Dim dbs As DAO.Database
    Dim tblDef As DAO.TableDef
    Dim fld As DAO.Field
    Dim idx As DAO.Index
    
    Set dbs = Application.CurrentDb
    Set tblDef = dbs.TableDefs("KundenPlusBestellung")
    
    Set idx = tblDef.CreateIndex("PrimaryKey")
    
    With idx
        Set fld = .CreateField("IDBestellsumme")
        .Fields.Append fld
        .Unique = False
        .Primary = True
    End With
    
    tblDef.Indexes.Append idx
    
    
    Set fld = Nothing
    Set idx = Nothing
    Set tblDef = Nothing
    Set dbs = Nothing
    
    Application.RefreshDatabaseWindow
End Sub


